
This paper describes three algorithms for q-hypergeometric summation:

• a multibasic analogue of Gosper’s algorithm,
• the q-Zeilberger algorithm, and

• an algorithm for finding q-hypergeometric solutions of linear recurrences

together with their Maple implementations, which is relevant both to people being

interested in symbolic computation and in q-series.

For all these algorithms, the theoretical background is already known and has been
described, so we give only short descriptions, and concentrate ourselves on introducing
our corresponding Maple implementations by examples. Each section is closed with a
description of the input/output specifications of the corresponding Maple command.

We present applications to q-analogues of classical orthogonal polynomials. In par-

ticular, the connection coefficients between families of q-Askey–Wilson polynomials are
computed.

Mathematica implementations have been developed for most of these algorithms,
whereas to our knowledge only Zeilberger’s algorithm has been implemented in Maple

so far (Koornwinder, 1993 or Zeilberger, cf. Petkovšek et al., 1996).

We made an effort to implement the algorithms as efficient as possible which in the

q-Petkovšek case led us to an approach with equivalence classes. Hence, our implemen-
tation is considerably faster than other ones. Furthermore the q-Gosper algorithm has
been generalized to also find formal power series solutions.

c© 1999 Academic Press

1. Introduction

The well-known algorithms of Gosper, Zeilberger and Petkovšek are useful tools for prob-
lems dealing with hypergeometric summation. The theory and a description of their im-
plementation in Maple are described in detail by Koepf (1998); the theory can also be
found in Petkovšek et al. (1996); for Maple details see also Koepf (1996).

The algorithms can be easily adapted to the q-case, see e.g., Koornwinder (1993) for
a short description of Gosper/Zeilberger and Abramov et al. (1998) for Petkovšek, and
Gasper and Rahman (1990) as a general reference for q-hypergeometric (basic) series.

Hence we sketch the underlying theory only briefly and put emphasis on the application
of our Maple package qsum.mpl which is an implementation of those algorithms. Our
purpose is to show how these procedures can be applied to solve, e.g., problems related
to orthogonal polynomials. Additionally we introduce a slight variation of the q-Gosper
algorithm that searches for formal power series solutions and a multibasic version that is
implemented in our package. To increase the usability of the programs we made an effort

0747–7171/99/120777 + 23 $30.00/0 c© 1999 Academic Press

to implement the algorithms as efficient as possible which in the q-Petkovšek case led
us to an approach with equivalence classes. Hence, our implementation is considerably
faster than other ones.

This package and an extensive help database for Maple V (Release 4 or 5) is avail-
able from the URL http://www.imn.htwk-leipzig.de/~koepf/research.html or by
e-mail request from koepf@imn.htwk-leipzig.de. All examples shown in this paper
were computed with Maple V Release 5 on a Pentium PC with 200 MHz. A corre-
sponding Maple worksheet with the computations of this paper and the timings can
also be obtained electronically from the above URL.

For Mathematica the package qZeil.m written by Riese is an excellent implementa-
tion of q-Gosper’s and q-Zeilberger’s algorithm.† Petkovšek implemented the q-Petkovšek
algorithm in Mathematica.‡ A Mathematica notebook with the examples of Table 2
and Section 4.1 using these packages is also available from the above URL.

Throughout this paper by q the vector (q1, q2, . . . , qm), m ∈ N (set of positive integers)
is denoted, and F is an abbreviation for K(q1, q2, . . . , qm), K denoting the field of rational
numbers Q extended by some parameters.§

2. qqq-Gosper

The q -version of Gosper’s algorithm determines for a q-hypergeometric term F(k), i.e.

F(k + 1)
F(k)

= H(k)

for all k ∈ Z (set of integers) with

H(k) ∈ F
(
qk1 , . . . , q

k
m

)
,

a q -hypergeometric term G(k) with

F(k) = G(k + 1)−G(k) (2.1)

iff it exists. Knowing such an antidifference¶ (also called indefinite sum) G(k) for F(k)
makes the evaluation of sums with given lower and upper bounds over F(k) easy:

h∑
k=l

F(k) = G(h+ 1)−G(l).

The q -Gosper algorithm is based on the following two observations:

(1) By dividing equation (2.1) by G(k) one can see that G(k)/F(k) ∈ F
(
qk1 , . . . , q

k
m

)
.

(2) On the other hand, if one divides equation (2.1) by F(k), one obtains

F(k + 1)
F(k)

G(k + 1)
F(k + 1)

− G(k)
F(k)

= 1,

†The package qZeil is available from http://www.risc.uni-linz.ac.at/research/combinat/risc/
software/qZeil.
‡The corresponding package qHyper.m (and the hypergeometric version Hyper.m) can be obtained at

http://www.mat.uni-lj.si/dwnld.htm.
§From a theoretical point of view any computable field of characteristic zero would be appropriate.

However, algorithms used in computer algebra systems (such as factorization) usually require a more
restrictive setting.
¶Note that by antidifference we always mean an upward antidifference.

http://www.imn.htwk-leipzig.de/~koepf/research.html
http://www.risc.uni-linz.ac.at/research/combinat/ risc/software/qZeil
http://www.risc.uni-linz.ac.at/research/combinat/ risc/software/qZeil
http://www.mat.uni-lj.si/dwnld.htm

or equivalently

H(k) C(k + 1)− C(k) = 1, C(k) =
G(k)
F(k)

∈ F
(
qk1 , . . . , q

k
m

)
. (2.2)

Thus to determine an antidifference G(k) for F(k) it is sufficient to search for a
rational solution C(k) of an inhomogeneous recurrence equation of first order with
rational coefficients.

As the q -Gosper algorithm is already well described by Koornwinder (1993) for the case
m = 1, by Riese (1996) for m = 2, and by Böing (1998) for the general case m ∈ N, we
merely want to give the two main steps of the algorithm without going into detail:

(1) Determine P(k),Q(k),R(k) ∈ F
[
qk1 , . . . , q

k
m

]
such that

H(k) =
F(k + 1)

F(k)
=

P(k + 1)
P(k)

Q(k)
R(k)

(2.3)

and
gcd(Q(k),R(k + j)) = 1, for all j ∈ N0 = N ∪ {0}. (2.4)

(Note that the polynomials P(k),Q(k),R(k) are not uniquely determined by these
two properties (cf. Lemma 4.1).)

(2) If the inhomogeneous first-order recurrence equation for X(k)

Q(k) X(k)− R(k − 1) X(k − 1) = P(k) (2.5)

has a solution X(k) ∈ F
[
qk1 , q

−k
1 , . . . , qkm, q

−k
m

]
, then

G(k) =
R(k − 1) X(k − 1)

P(k)
F(k) (2.6)

is a q -hypergeometric antidifference for F(k). Otherwise no such antidifference ex-
ists.

With the q -Gosper algorithm one can derive identities, like e.g., Formula (II.34) found
in the Appendix of Gasper and Rahman (1990):

n∑
k=0

1− a pk qk

(1− a) ck
(a; p)k

(a p/c; p)k

(c; q)k
(q; q)k

=
1
cn

(a p; p)n (c q; q)n
(a p/c; p)n (q; q)n

, (2.7)

where (a; q)k denotes the q-Pochhammer symbol defined as usual by

(a; q)k =

 (1− a)(1− a q) · · · (1− a qk−1), if k > 0,
1, if k = 0,
[(1− a q−1)(1− a q−2) · · · (1− a qk)]−1, if k < 0.

(2.8)

After loading our package in Maple via

> read ‘qsum.mpl‘;

Copyright (c) 1998, Harald Böing & Wolfram Koepf

one can deduce the right-hand side of equation (2.7) from the left-hand side with the
procedure qgosper:

> result:= qgosper((1-a*p^k*q^k)/(1-a)/c^k*qpochhammer(a,p,k)/
qpochhammer(a*p/c,p,k)*qpochhammer(c,q,k)/qpochhammer(q,q,k),[p,q],k=0..n);

result := qpochhammer(c, q, n+ 1) qpochhammer(a, p, n+ 1)(−1 + q(n+1))

(−c+ p(n+1) a)c(−n−1)
/

(

qpochhammer(q, q, n+ 1)qpochhammer
(
a p

c
, p, n+ 1

)
(−1 + a)(−1 + c)).

It is easy to check that this result is equivalent to the right-hand side of equation (2.7),
e.g., by the computation

> qsimpcomb(result/qpochhammer(a*p,p,n)/qpochhammer(c*q,q,n)*
qpochhammer(a*p/c,p,n)*qpochhammer(q,q,n)*c^n);

1

The procedure qsimpcomb is a simplification procedure for q -hypergeometric terms. It
was designed to simplify ratios of q -hypergeometric terms like F(k + 1)/F(k) to decide
rationality. If the result is nonrational, the procedure qsimplify returns a more compact
result than qsimpcomb.

Generally the most time-consuming part in q -Gosper’s algorithm is its second step,
which is usually done in two parts:†

(1) Determine lower and upper degree bounds l and h for X(k).
(2) Substitute a generic Laurent polynomial

∑h
i=l ci (qk)i for X(k) with as yet unknown

coefficients ci in equation (2.5) and solve the resulting linear system by comparing
the coefficients of (qk)i.

This method involves solving a linear system in h−l+1 variables which is inefficient if the
difference h− l is large. Abramov et al. (1995) introduced an alternative algorithm. It is
based on the idea to convert the recurrence equation for X(k) into one for the coefficients
ci and using this recurrence to calculate as many coefficients as possible. Abramov et al.
(1995) suggested using their algorithm if the difference h − l is greater or equal to the
order of the recurrence equation that is to be solved.

We want to illustrate this by calculating an antidifference for the function

Fn(k) =
(a; q)k
(q; q)k

(a qn)−k, (2.9)

where a is an arbitrary parameter. The application of q-Gosper’s algorithm to Fn(k) with
symbolic n shows that no q-hypergeometric antidifference exists:

†To simplify the notation, we just describe the case m = 1.

> qgosper(qpochhammer(a,q,k)/qpochhammer(q,q,k)/(a*q^n)^k,q,k);

Error, (in qgosper) No q-hypergeometric antidifference exists.

However, for each nonnegative integer n ∈ N0 there exists an antidifference, where the
resulting recurrence equation for X(k), i.e.

(1− a qk) X(k)− a qn(1− qk)X(k − 1) = 1, (2.10)

implies X(k) being a polynomial of degree n. Thus for n = 1 we obtain, e.g.,footnoteBy
default the resulting antidifference is factorized over F[qk], which can be rather time-
consuming if X(k) is ‘complicated’. The application of this simplification can be dropped
via the option simplify=false.

> qgosper(qpochhammer(a,q,k)/qpochhammer(q,q,k)/(a*q)^k,q,k,simplify=false);

a q (−1 + qk)
(

1
−1 + a q

+
a (q − 1)qk

(−1 + a q)(−1 + a) q

)
qpochhammer(a, q, k)

(a q)k qpochhammer(q, q, k)
.

The system of linear equations can be solved by qgosper† with three different methods
that are invoked via the optional argument “solvemethod=name”, where name is one of

ABP: This represents the algorithm of Abramov et al. (1995) mentioned above that we
implemented for m = 1.

solve: This is Maple’s builtin solve procedure.
gausselim: An implementation of Gaussian elimination, since Maple’s solve in many

cases only poorly solves systems of linear equations. The same problem occurs in
Mathematica and was described by Paule and Riese (1997).

auto: This is the default, meaning that qgosper chooses either ABP or gausselim, de-
pending on the example. In most cases it chooses the faster method.

Table 1 shows the timings for computing antidifferences of Fn(k), generated by the call

qgosper(qpochhammer(a,q,k)/qpochhammer(q,q,k)/(a*q^n)^k,q,k,simplify=false);

with specified n, and the additional optional argument solvemethod set to the appro-
priate value:‡

In Section 4 we will introduce an extension of the one-dimensional q-Gosper algorithm
that allows the computation of an antidifference for Fn(k) for arbitrary n ∈ N0.

Note that—similarly to the process of indefinite integration—an antidifference is only
determined up to an additive constant, as the following example for m = 3 shows. The
function qbrackets(k,q) is given as

[k]q =
qk − 1
q − 1

.

† This also applies to qsumrecursion, introduced in the next section.
‡ The occurring dashes (—) in the table mean that no result was obtained within one hour.

Table 1. Timings for different solve methods.

n 10 20 30 40 50 60 70 80 90 100

solve 6 s 128 s — — — — — — — —
gausselim 1 s 3 s 8 s 19 s 39 s 76 s 136 s 210 s 294 s 425 s
ABP 0 s 1 s 3 s 9 s 23 s 48 s 86 s 163 s 199 s 275 s

We define t(k) by

> t:= qbrackets(k,p)*qbrackets(k,q)*qbrackets(k,r):

Obviously t(k) is an antidifference of t(k + 1)− t(k). Let us nevertheless compute an
antidifference of t(k + 1)− t(k) applying qgosper:

> result:= qsimpcomb(qgosper(subs(k=k+1,t)-t,[p,q,r],k));

result :=
rk + qk − qk rk + pk − pk rk − pk qk + pk qk rk

(p− 1)(q − 1)(r − 1)
.

With qsimpcomb one shows that this antidifference is the same as t with an additive
constant (w.r.t. k):

> qsimpcomb(result-t);

1
(p− 1)(q − 1)(r − 1)

.

2.1. description of the Maple procedure qgosper

A thorough reference of the procedures in the package qsum.mpl is accessible in Maple

V (Release 4 or 5) via the accompanying help file maple.hdb.† For a general introduction
to the package one should type ?qsum in Maple. To obtain help for a special procedure
one can either follow the given links or use e.g., ?qgosper directly in Maple.

Some of the procedures allow certain optional arguments. These can be specified in an
arbitrary order after the regular arguments and are always of the type optname=optvalue
where optname is the name of the option and optvalue a possible value for this option.

Calling Sequence:

qgosper(f,q,k,...);
qgosper(f,[q1,q2,...],k,...);
qgosper(f,q,k=l..h,...);

†To have access to the help database, set libname:=‘libname‘,‘directory‘. Here directory denotes
the directory where maple.hdb resides.

Parameters:
f — an algebraic expression, typically a product/ratio of

rational functions, powers, qpochhammer’s, qbinomial’s,
qfactorial’s, qbrackets’ and qphihyperterm’s, compare
Section 3.

k — a name, the summation variable
q,q1,q2,... — names
l,h — algebraic expressions, the summation range
... — a sequence of optional arguments (of type name=name)

Input/Output Description:
This function decides whether for a q -hypergeometric term f(k) a q -hypergeometric
antidifference g(k) exists. If such an antidifference exists, it will be returned; oth-
erwise an error message is generated stating that no such antidifference exists. If
the input is not q -hypergeometric w.r.t. k, the function will return an appropriate
error message.

Options and Remarks:
• If qgosper is called by qgosper(f,q,k=l..h), it will return a term that is equal

to sum(f,k=l..h) if a q -hypergeometric antidifference for f exists.
• If qgosper returns no antidifference, one might try the option series=true, thus

allowing qgosper to search for an antidifference that is a q-hypergeometric series,
see Section 4.3.

• By default qgosper returns an upward antidifference. This can be changed by using
the option antidifference=down.

• Different solve methods can be accessed via the option solvemethod=method where
method is one of ABP, gausselim, or solve.

• If the resulting antidifference is complicated, it might be useful to avoid any sim-
plifications; this can be achieved via simplify=false. By default the antidif-
ference is factorized. If you want to apply simpfunc to the result instead, use
simplify=simpfunc.

3. q-Zeilberger

Wilf and Zeilberger (1992) showed that Zeilberger’s algorithm can be easily carried
over to the q-case (see also the description by Koornwinder, 1993). The q-Zeilberger
algorithm tries to derive a recurrence equation for the definite sum S(n)

S(n) =
bn∑

k=an

F(n, k), (3.1)

where F(n, k) is a q-hypergeometric term w.r.t. n and k. It uses the q-Gosper algorithm
to determine an antidifference G(n, k) for fn(k) and σ0(n), . . . , σJ (n) ∈ F(qn) for some
J ∈ N, where†

fn(k) =
J∑
j=0

σj(n) F(n− j, k). (3.2)

† Instead of F(n − j, k) one could also use F(n + j, k) leading to an upward instead of a downward
recurrence equation.

Note that q-Gosper’s algorithm can be applied to fn(k) since it is a rational multiple
of F(n, k). The application of q-Gosper to fn(k) in the second step gives a recurrence
equation for X(k) (see equation (2.5)) which leads to a system of equations that are
linear in the coefficients of X(k) and in σ0(n), . . . , σJ(n). Thus the only change required
is to add the unknowns σ0(n), . . . , σJ(n) to the variables of the linear system that has to
be solved.

If the algorithm is successful, we obtain an inhomogeneous recurrence equation for
F(n, k),

J∑
j=0

σj(n) F(n− j, k) = G(n, k + 1)−G(n, k), (3.3)

with σ0(n), . . . , σJ (n) ∈ F(qn). By summing over k from αn = max{an−j | j = 0, . . . , J}
to βn = min{bn−j | j = 0, . . . , J} we get the inhomogeneous recurrence equation

J∑
j=0

σj(n) S(n− j) = G(n, βn + 1)−G(n, αn) +

J∑
j=0

[
αn−1∑
k=an−j

F(n− j, k) +
bn−j∑

k=βn+1

F(n− j, k)

]
(3.4)

for S(n).
An advantage of Zeilberger’s method is that, no matter how complicated the compu-

tation of the recurrence equation (3.4) might be, one can prove its validity by rational
arithmetic if one also outputs G(n, k) or the ratio G(n, k)/F(n, k):

(1) Write down the corresponding recurrence equation (3.3) for the summand F(n, k)
and divide it by F(n, k). As G(n, k)/F(n, k) ∈ F

(
qn, qk

)
, this equation can be shown

by pure rational arithmetic.†
(2) Knowing that equation (3.3) is correct, we proceed by summing over k to obtain

equation (3.4).

Thus the term G(n, k)/F(n, k) is often referred to as the (rational) certificate for the
recurrence equation (3.4). Since the term F(n, k) usually has finite support, i.e. F(n, k)
vanishes for each n ∈ N0 outside a finite interval of k, in most cases the right-hand side
of (3.4) is zero. Therefore our implementation assumes that the right-hand side of (3.4)
is zero, if no summation bounds are specified (thus saving computation time).
Suppose we want to prove the q-Pfaff–Saalschütz identity

3φ2

(
a, b, q−n

c, a b q1−n/c

∣∣∣∣ q; q) =
(c/a; q)n (c/b; q)n
(c; q)n (c/(a b); q)n

, (n ∈ N0) (3.5)

where the basic hypergeometric series rφs is defined by

rφs

(
a1, a2, . . . , ar
b1, b2, . . . , bs

∣∣∣∣ q; z) =
∞∑
k=0

(a1, a2, . . . , ar; q)k
(b1, b2, . . . , bs; q)k

zk

(q; q)k

(
(−1)k q(

k
2)
)1+s−r

, (3.6)

and (a1, a2, . . . , ar; q)k is an abbreviation for
∏r
j=1 (aj ; q)k. With our implementation of

the q-Zeilberger algorithm (qsumrecursion) one obtains:

†After simplifying the ratios of the q-hypergeometric terms.

> qsumrecursion([a,b,q^(-n)],[c,a*b*q^(1-n)/c],q,q,S(n),
rec2qhyper=true,sumrange=0..n);

S(n) =
qpochhammer

( c
a
, q, n

)
qpochhammer

(c
b
, q, n

)
qpochhammer(c, q, n) qpochhammer

( c

b a
, q, n

) , 0 ≤ n

 .
To enforce the computation of the inhomogeneous part† we supplied the optional argu-
ment sumrange=0..n; the option rec2qhyper=true advises qsumrecursion to return a
q-hypergeometric term instead of a recurrence equation if the order is one.

Additionally to the output, the procedure stored some information in the global vari-
able _qsumrecursion_proof which is a table with several entries. One obtains, e.g., the
input term, the recurrence equation and the certificate G(n, k)/F(n, k) for (3.3) by

> term:= _qsumrecursion_proof[_F];

term :=
qpochhammer(a, q, k) qpochhammer(b, q, k) qpochhammer

(
q−n, q, k

)
qk

qpochhammer(c, q, k) qpochhammer
(
a b q1−n

c
, q, k

)
qpochhammer(q, q, k)

> RE:= _qsumrecursion_proof[_recursion];

RE := (qn c− q) (−qn c+ a b q)S(n) + (−qn c+ b q) (a q − qn c) S(n− 1) = 0

> cert:= _qsumrecursion_proof[_certificate];

cert := − (−qn c+ a b q) (−q + c qk) (−1 + qk) qn

(qn − 1) qk
.

Due to the rφs-series type of input qsumrecursion had to choose a summation variable
(namely k) such that S(n) is a sum of term over k from 0 to n. However, be careful to
avoid interference with the global variable k we defined k to be local.‡ The best way to
use this local variable (stored in _qsumrecursion_proof[_sumvar]) is to assign it to a
global variable, e.g.

> i:=_qsumrecursion_proof[_sumvar];

i := k

One can now prove the recurrence equation (3.3) for the summand F(n, k) (term) by
rational arithmetic, where the antidifference G(n, k) is F(n, k)G(n,k)

F(n,k) , hence cert*term:

† As we expected, it turned out to be zero.
‡ In other words, a substitution, such as subs(k=k+1,term) would not work!

> result:=subs({S(n-1)=subs(n=n-1,term),S(n)=term},lhs(RE))-
(subs(i=i+1,cert*term)-cert*term):

> qsimpcomb(result);

0

Therefore equation (3.3) is valid, which can now be converted into a recurrence equation
for S(n) by summing over k: because F(n, k) and the antidifference G(n, k) have finite
support, i.e.

(a; q)k (b; q)k (q−n; q)k q
k

(c; q)k (a b q1−n/c; q)k (q; q)k
= 0, for k < 0 or k > n

the easiest way to prove _qsumrecursion_proof[_recursion] now is by summing over
k from −∞ to ∞, showing that the inhomogeneous part is zero.

Next let us apply our program to deduce the right-hand side of the q-Dixon identity

n∑
k=−n

(−1)k qk (3 k−1)/2

[
n+ b

n+ k

]
q

[
n+ c

c+ k

]
q

[
b+ c

b+ k

]
q

=
(q; q)n+b+c

(q; q)n (q; q)b (q; q)c
, (3.7)

where the q-binomial coefficient is defined by[n
k

]
q

=
(q; q)n

(q; q)k (q; q)n−k
, (n ∈ N0).

First we note that the command sum2qhyper can be used to convert q-hypergeometric se-
ries into q-hypergeometric notation, hereby unmasking their lower and upper parameters
as well as their argument; we obtain e.g., for the q-Dixon sum:

> term:= (-1)^k*q^(k*(3*k-1)/2)*qbinomial(n+b,n+k,q)*qbinomial(n+c,c+k,q)*
qbinomial(b+c,b+k,q):

> assume(n,integer):
sum2qhyper(term,q,k);
n:=’n’:

(−1)(−n) q(1/2n+3/2n2) qpochhammer(q, q, b+ c)
φ([q(−2n), q(−n−b), q(−c−n)], [q(1−n+b), q(1−n+c)], q, qn+b+c+1)/(
qpochhammer(q, q, c− n) qpochhammer(q, q, 2n) qpochhammer(q, q, b− n)).

The assumption that n is an integer was made so that the program can compute the
appropriate shift to let the series start with k = 0. The upper parameters of the resulting
expression show in particular that the sum consists of 2n+ 1 summands (if neither b nor
c are integers smaller than n).

If we apply qsumrecursion to find a recurrence equation for the sum, we obtain

> qDixon_RE:= qsumrecursion(term,q,k,S(n));

qDixon RE :=
(
− q2n + q

)(
qn + 1

)(
qn − 1

)
q3 S(n) + q

(
q5 + q4 − q4+n+c+b −

q2n+3 − qn+3+c − qn+3+c+b + q3 + q2n+3+c + q2n+3+b − qn+3+b − q3n+2 +
q2n+2+b + q4n+2+c+b + q2n+2+c − q3n+2+b − q3n+2+c − q2n+2+c+b − q1+3n +
q4n+1+c+b + q4n+c+b

)
S(n− 1)−

(
q − qn+c+b

)(
− q3n+c+b+1 + q5 + q2n+2 −

q3n+c+b + q4 − q3n+2+c+b + q6 − q4+n+c+b + q2n+3+b − qn+3+c + q2n+3+c −
qn+3+b − q4+n+c + q2n+2+b + q2n+2+c − q4+n+b

)
S(n− 2) +(

q2 − qn+c+b
)(
q − qn+c+b

)(
q2 − qn+b

)(
q2 − qn+c

)
S(n− 3) = 0, (3.8)

which is definitely not the result we want. Here we have an example (of quite a few),
where the q-Zeilberger algorithm does not find a recurrence equation of minimal order,
which was pointed out by Paule and Riese (1997). Paule (1994) introduced the method
of creative symmetrizing—a generalization of which is given by the following lemma—
resolving the problem of non-minimality in most cases.

Lemma 3.1. (Riese, 1995) If for some c ∈ Z
bn∑

k=an

F(n, k) =
bn∑

k=an

F(n,−k − c),

and M(n, k) = F(n,−k − c)/F(n, k) then

bn∑
k=an

F(n, k) =
1
2

bn∑
k=an

(
1 + M(n, k)

)
F(n, k).

The astonishing fact is that, in most cases, the q-Zeilberger algorithm applied to the sym-
metrized summand 1/2 (1 + M(n, k)) F(n, k) increases the chance of getting a recurrence
equation of minimal order. For q-Dixon we choose c = 0 to obtain:

> M:= qsimpcomb(subs(k=-k,term)/term);

M := qk

If we now apply qsumrecursion to the symmetrized summand, we obtain:

> qsumrecursion((1+M)/2*term,q,k,S(n));

(
qn − 1

)
S(n) −

(
− 1 + qn+b+c

)
S(n− 1) = 0.

Table 2 is an extract of the one given in Paule and Riese (1997) with the most time-
consuming examples, where we used the newest version of their package qZeil (Version
1.8) with Mathematica 3.0.

Turbo-qZeil just means that infinite summation bounds were specified, thus disabling
the computation of the inhomogeneous part. Our procedure did not compute the inho-
mogeneous part either, but instead of specifying the order of the recurrence (which is



788 H. Böing and W. Koepf

Table 2. Comparison of the procedures qsumrecursion and qZeil.

Equation† Order qsumrecursion Turbo-qZeil

lhs 2 2 s 4 s(III.18)
rhs 2 1 s 5 s

lhs 3 4 s 11 s(III.25)
rhs 3 13 s 27 s

lhs 2 17 s 31 s(III.28)
rhs 2 20 s 39 s

necessary for qZeil), qsumrecursion tries to find a recurrence from first order up to the
one, where it is successful. By default it looks for a recurrence of order one up to five, as
in most cases the computing time of the non-successful tries is not crucial.

Note that an earlier implementation of qZeil failed with some of these examples, see
Paule and Riese (1997).

The q-Zeilberger algorithm also gives a very simple method to deduce recurrence equa-
tions for orthogonal polynomials from the basic Askey–Wilson tableau (see e.g. Koekoek
and Swarttouw, 1994). For the little q-Legendre polynomials

pn(x|q) = 2φ1

(
q−n, qn+1

q

∣∣∣∣ q; qx)
we get e.g., the recurrence equations

> qsumrecursion([q^(-n),q^(n+1)],[q],q,q*x,p(n));

qn (qn − 1) (q + qn) p(n) + (2 qn − x qn+1 − q x− x qn − x q2n) (q − q2n) p(n− 1)−
qn (qn + 1) (q − qn) p(n− 2) = 0

with respect to n, and

> qsumrecursion([q^(-n),q^(n+1)],[q],q,q*q^y,p(y));

qn+1 (qy − 1) p(y) − (qy + qy+2n+1 − 2 qn+1) p(y − 1) + qn (qy − q)p(y − 2) = 0

with respect to y. One can also compute the q-difference equation for the little
q-Legendre polynomials

> qsumdiffeq([q^(-n),q^(n+1)],[q],q,q*x,p(x));

q (qn − 1) (qn q − 1) p(x) + (−1 + q) (−qn + qn q − q3 x qn + q2 x (qn)2 + q x− q2 x qn)·
Dqx(p(x))− qn (−1 + q2 x)x (−1 + q)2 qDqx, x(p(x)) = 0,

†All equation numbers refer to the Appendix of the book of Gasper and Rahman (1990).

Dqx denoting the q-derivative operator

Dqx f(x) :=
f(x)− f(qx)

(1− q)x
.

Furthermore, the computation

> qsumrecursion([q^(-n),q^(n+1)],[q],q,q,p(n),rec2qhyper=true);

[
p(n) = (−q)n qbinomial(n,2), 0 ≤ n

]
shows that

pn(1|q) = (−1)n qn (n+1)/2.

Now consider the Al–Salam–Chihara polynomials with x = cos θ, given by:

Qn(x; a, b|q) =
(a b; q)n
an

3φ2

(
q−n, a ei θ, a e−i θ

a b, 0

∣∣∣∣ q; q) (3.9)

=
(
b e−i θ; q

)
n
ei n θ 2φ1

(
q−n, a ei θ

b−1 q1−n ei θ

∣∣∣∣ q; b−1 q e−i θ
)
. (3.10)

With qsumrecursion one can easily show the equality of the two representations by

> re1:= qsumrecursion(qpochhammer(a*b,q,n)/a^n,
[q^(-n),a*exp(I*theta),a*exp(-I*theta)],[a*b,0],q,q,Q(n));

−eI θ q3 Q(n) − (eI θ qn a− q + eI θ qn b− q e2 I θ) q2 Q(n− 1) +
eI θ (−q2 + a b qn) (−qn + q) Q(n− 2) = 0

> re2:= qsumrecursion(qpochhammer(b*exp(-I*theta),q,n)*exp(I*n*theta),
[q^(-n),a*exp(I*theta)],[b^(-1)*q^(1-n)*exp(I*theta)],
q,b^(-1)*q*exp(-I*theta),Q(n));

−eI θ q3 Q(n) − (eI θ qn a− q + eI θ qn b− q e2 I θ) q2 Q(n− 1) +
eI θ (−q2 + a b qn) (−qn + q) Q(n− 2) = 0.

Since the two recurrence equations agree, (3.9) and (3.10) really define the same poly-
nomials as long as the two initial values Q0(x; a, b|q) and Q1(x; a, b|q) are shown to be
equal for both representations.†

Finally we consider the Askey–Wilson polynomials pn(x; a, b, c, d|q) with x = cos θ,
given by

pn(x; a, b, c, d|q) = (a b, a c, a d; q)n a
−n

† Note that the Riese implementation does not allow this type of input and returns the error message:
a ei θ is not a valid power product. With this package you have to replace ei θ by y, apply the procedure
qZeil, and use the backsubstitution y = ei θ and the Simplify command.

4φ3

(
q−n, a b c d qn−1, a ei θ, a e−i θ

a b, a c, a d

∣∣∣∣ q; q) (3.11)

which are on top of the basic Askey–Wilson tableau (see e.g. Koekoek and Swarttouw,
1994). All the other basic orthogonal polynomial families in this tableau can be con-
structed from them by limiting procedures.

Gasper and Rahman (1990, Sections 7.5 and 7.6), considered the connection coefficients
ck(n) between two Askey-Wilson polynomials with different parameters

pn(x;α, β, γ, d|q) =
n∑
k=0

ck(n;α, β, γ, a, b, c, d) pk(x; a, b, c, d|q). (3.12)

This equation expresses the polynomials of one family by a linear combination of poly-
nomials of another family.

Askey and Wilson (1985) showed the following representation for ck(n)

ck(n;α, β, γ, a, b, c, d) =
(αd, β d, γ d, q; q)n

(
αβ γ d qn−1; q

)
k

(αd, β d, γ d, q, a b c d qk−1; q)k (q; q)n−k
qk

2−nk dk−n

5φ4

(
qk−n, α β γ d qn+k−1, a d qk, b d qk, c d qk

a b c d q2 k, α d qk, β d qk, γ d qk

∣∣∣∣ q; q) (3.13)

in terms of a basic hypergeometric function.
For the special case β = b and γ = c Gasper and Rahman (1990) gave the formula

ck(n;α, b, c, a, b, c, d) =
(α/a; q)n−k

(
α b c d qn−1; q

)
k

(q, b c, b d, c d; q)n a
n−k

(q, b c, b d, c d; q)k (a b c d qk−1; q)k (q, a b c d q2 k; q)n−k
(3.14)

as a q-hypergeometric term.
Note that in their book this equation—which is number (7.6.9)—contains a misprint,

discovered by our software, i.e. the factor (q; q)n is missing (cf. Gasper and Rahman,
1997).

From equation (3.13) with β = b and γ = c our implementation detects a recur-
rence equation of first order for ck(n), from which the resulting q-hypergeometric term
is computed using an initial value for k = n:

> c:=(k,n,alpha,beta,gamma,a,b,c,d,j)->
qpochhammer(alpha*d,beta*d,gamma*d,q,q,n)*
qpochhammer(alpha*beta*gamma*d*q^(n-1),q,k)/
qpochhammer(alpha*d,beta*d,gamma*d,q,a*b*c*d*q^(k-1),q,k)/
qpochhammer(q,q,n-k)*q^(k^2-n*k)*d^(k-n)*
qphihyperterm([q^(k-n),alpha*beta*gamma*d*q^(n+k-1),a*d*q^k,b*d*q^k,c*d*q^k],
[a*b*c*d*q^(2*k),alpha*d*q^k,beta*d*q^k,gamma*d*q^k],q,q,j):

> intermediate:=qsumrecursion(c(k,N+k,alpha,b,c,a,b,c,d,j),q,j,S(N),
rec2qhyper=true,sumrange=0..N):
final:= subs(N=n-k,rhs(intermediate[1]));

final := ((−q + b d qk α c) qpochhammer(b d qk α c, q, k) (−q + a b c d (qk)2)

qpochhammer(b c qk, q, n− k) qpochhammer
(
α b c d (qk)2

q
, q, n− k

)

qpochhammer
(
α

a
, q, n− k

)
qpochhammer(q qk, q, n− k)

qpochhammer(c d qk, q, n− k) qpochhammer(b d qk, q, n− k) a(n−k))
/

(

(−q + α b c d (qk)2) (−q + a b c d qk) qpochhammer(a b c d qk, q, k)
qpochhammer(q, q, n− k) qpochhammer(a b c d (qk)2, q, n− k)

qpochhammer
(
b d qk α c

q
, q, n− k)

)
which gives the right-hand side of equation (3.14).† This can be easily checked by the
computation

> RHS:=qpochhammer(alpha/a,q,n-k)*qpochhammer(alpha*b*c*d*q^(n-1),q,k)*
qpochhammer(b*c,b*d,c*d,q,n)*a^(n-k)*qpochhammer(q,q,n)/
(qpochhammer(q,b*c,b*d,c*d,q,k)*qpochhammer(a*b*c*d*q^(k-1),q,k)*
qpochhammer(q,a*b*c*d*q^(2*k),q,n-k)):
qsimpcomb(final/RHS);

1

Note that similar computations with just one differing parameter—i.e. α = a, γ = c
or α = a, β = b—also lead to recurrence equations of order one and hence to q-
hypergeometric terms. These results are given by

ck(n; a, β, c, d, a, b, c, d) =
(−q + d β c qk a) (−q + a b c d q2k) bn−k

(−q + d β c q2k a) (−q + a b c d qk)

(
d β c qk a; q

)
k

(a b c d qk; q)k

×
(
a c qk, d β c q2k−1 a, β/b, qk+1, c d qk, a d qk; q

)
n−k

(q, a b c d q2k, d β c qk−1 a; q)n−k
and

ck(n; a, b, γ, d, a, b, c, d) =
(−q + d γ qk a b) (−q + a b c d q2k) cn−k

(−q + a b γ d q2k) (−q + a b c d qk)

(
d γ qk a b; q

)
k

(a b c d qk; q)k

×
(
a b qk, a b γ d q2k−1, γ/c, qk+1, b d qk, a d qk; q

)
n−k

(q, a b c d q2k, d γ qk−1 a b; q)n−k
from which one can derive the connection coefficients between many families of the q-
Askey–Wilson tableau by limit computations. By the symmetry of equation (3.11), a
similar connection is valid if the fourth parameter d is varied.

3.1. description of the Maple procedure qsumrecursion

Calling Sequence:

qsumrecursion(f,q,k,s(n),...);
qsumrecursion(f,q,k=l..h,s(n),...);

† The procedure qphihyperterm returns the summand of the basic hypergeometric series (3.6).

qsumrecursion(f,upper,lower,qq,z,s(n),...);

Parameters:
f — an algebraic expression
q — a name
k — a name, the summation variable
s — a name, the recurrence function
n — a name, the recurrence variable
l,h — algebraic expressions, lower and upper summation bounds
upper, lower — lists of algebraic expressions, the upper and lower parame-

ters of the general q-hypergeometric function
qq — a name or a nameinteger

... — a sequence of optional arguments (of type name=name)

Input/Output Description:
qsumrecursion applies the q-Zeilberger algorithm to determine a recurrence equa-

tion for the sum s(n) =
h∑
k=l

f(n, k). If successful, the recurrence equation will be

returned; otherwise an error message is generated.

Options and Remarks:
• The last calling sequence is a shortcut for
qsumrecursion(f*qphihyperterm(upper,lower,qq,z,j),q,j,s(n));
where the prefactor f can be omitted if it is equal to one.

• The option recursion may be set to up or down, instructing qsumrecursion to
return an upward or downward recurrence equation, respectively.

• Similarly as in qgosper the option solvemethod may be set.
• If no explicit summation range is given, then it is assumed that the bounds are
-infinity..infinity, and a homogeneous recurrence equation is computed. If
other bounds are given, then qsumrecursion computes the inhomogeneous part of
the recurrence.
• Additionally one can supply the argument inhomo2homo=true; then qsumrecursion

will convert a resulting inhomogeneous recurrence equation of order J into a ho-
mogeneous one of order J + 1 if possible.

• By default qsumrecursion calculates an upward certificate for the recurrence equa-
tion; this may be changed by certificate=down.

• The option rec2qhyper=true tells the procedure to return a list with two entries if
a recurrence equation of first order was found: the first entry is a q-hypergeometric
term equal to the sum s(n), the second entry a restriction on n.

• The option proof can be set to any (unassigned) name. Then all necessary informa-
tion needed to prove the recurrence equation will be stored as a table in that name;
by default this information is stored in the global variable _qsumrecursion_proof.

4. qqq-Petkovšek

Abramov et al. (1998) gave a few algorithms to find solutions of special types of homo-
geneous and inhomogeneous recurrence equations with polynomial coefficients. Petkovšek
implemented the procedure qHyper in Mathematica that finds q-hypergeometric solu-
tions of homogeneous recurrence equations.

4.1. q-hypergeometric solutions of homogeneous recursions

Assume that we want to determine all q-hypergeometric term solutions F(n) of the
homogeneous recurrence equation

J∑
j=0

σj(n) F(n+ j) = 0, σ0(n), . . . , σJ(n) ∈ F[qn] . (4.1)

As in Gosper’s algorithm (see equation (2.3)), the following q-Gosper–Petkovšek repre-
sentation of rational functions plays a fundamental role:

Lemma 4.1. For every nonzero rational function L(n) ∈ F(qn) there exists a unique
quadruple (z,P(n),Q(n),R(n)) ∈ (F×F[qn]3), the q-Gosper–Petkovšek representation of
L(n), with

(i) L(n) = z
P(n+ 1)

P(n)
Q(n)
R(n)

,

(ii) gcd(Q(n),R(n+ j)) = 1, for all j ∈ N0,

(iii) gcd(Q(n),P(n)) = 1, and gcd(R(n),P(n+ 1)) = 1,

(iv) P(n),Q(n), and R(n) are monic,† and ldeg(P(n)) = 0.

After dividing equation (4.1) by F(n), substituting F(n + 1)/F(n) by its corresponding
q-Gosper–Petkovšek representation and clearing common denominators one obtains:

J∑
j=0

zj Cj(n) P(n+ j) = 0, (4.2)

where

Cj(n) :=
∞∑
k=0

cj,k(qn)k := σj(n)

(
j−1∏
i=0

Q(n+ i)

)(
J−1∏
i=j

R(n+ i)

)
∈ F[qn] .

Using Lemma 4.1 it is easy to deduce the two conditions

Q(n) divides σ0(n), and R(n) divides σJ (n− J + 1),

from the above equation. Besides, one can derive
J∑
j=0

cj,λ z
j = 0, λ = min{ldeg(Cj(n))

∣∣ j = 0, . . . , J} (4.3)

by equating the coefficients of (qn)λ in (4.2).
Now the q-Petkovšek algorithm for each possible choice† of Q(n), R(n) computes pos-

sible values of z, i.e. the set

Z =
{
x ∈ F

∣∣∣ ∑J

j=0
cj,λ x

j = 0
}
.

†P(n) is called monic if the leading coefficient is one.
†Possible choices for Q(n), e.g., are all combinations of non-constant monic divisors (w.r.t. n) of σ0(n).

Since all divisors can be assumed to be monic, there are only finitely many choices.

Once we know these, we substitute them into equation (4.2) and check whether there is
a polynomial solution P(n).

This means that the algorithm is rather slow if the coefficients σ0(n) and σJ(n) have
many factors, as we have to try a lot of combinations. Thus to apply the algorithm, it is
really important to have an efficient implementation.

We observed that the number of computations could be substantially decreased by
grouping the elements of Z into equivalence classes according to

x ∼ x̃ ⇐⇒ x = qi x̃ for some i ∈ Z.

Assuming that F(n) is a solution of equation (4.1), (z,P(n),Q(n),R(n)) the q-Gosper–
Petkovšek representation of F(n+ 1)/F(n), and z = z̃ qi for some integer i, we obtain

z
P(k + 1)

P(k)
Q(k)
R(k)

= z̃
P̃(k + 1)

P̃(k)
Q(k)
R(k)

, with P̃(n) =
(
qk
)i P(n).

This means that instead of searching for polynomial solutions P(n) of (4.2) for every
z ∈ Z it suffices to search for all Laurent polynomial solutions P(n) for one representative
of each equivalence class in Z. By dividing Z into equivalence classes the computation
times usually are halved.

Our implementation qrecsolve finds all q-hypergeometric solutions of recurrence equa-
tion (3.8) that was stored in the variable qDixon_RE, after 146 s:†

> Sn:= qrecsolve(qDixon_RE,q,S(n),return=qhypergeometric);

Sn :=

[[
qpochhammer

(
qc+1+b, q, n

)
qpochhammer(q, q, n)

, 0 <= n

]]
.

The optional argument return=qhypergeometric advises the procedure to return the
q-hypergeometric term S(n) and not the ratio S(n+1)/S(n) which is the default. Modulo
a constant factor w. r. t. n this should be equivalent to the right-hand side of (3.7):

> qsimplify(Sn[1][1]*qpochhammer(q,q,n)*qpochhammer(q,q,b)*
qpochhammer(q,q,c)/qpochhammer(q,q,n+b+c));

(−1 + qb) qpochhammer(q, q, c)
(−1 + qb+c) qpochhammer

(
qb, q, c

) .
Since Q(n) and R(n) are divisors of σ0(n) and σJ(n−J + 1), respectively, we can choose
to try only possible combinations of factors of the σ’s obtained by rational or by a
complete factorization. Of course the latter method is eventually slower, but we are sure
to obtain all solutions. If we are not interested in solutions over any extension field,
however, we might prefer rational factorization. This can be accomplished by using the
additional argument split=false in qrecsolve. If we do this for the q-Dixon example,
the solution is already found after 42 s.†

†With Petkovšek’s implementation qHyper (version from July 1995) and Mathematica 3.0 no result
was obtained after one hour. Note that we had to supply the optional arguments Solutions->All and
Quadratics->True to obtain a behavior equivalent to our procedure qrecsolve.
† Without Quadratics->True Petkovšek’s implementation needs 290 s.

4.2. q-hypergeometric solutions of inhomogeneous recursions

From the computational point of view the case of an inhomogeneous recurrence equa-
tion is much easier. If F(n) is a q-hypergeometric solution of the recurrence equation

J∑
j=0

σj(n) F(n+ j) = G(n), σ0(n), . . . , σJ (n) ∈ F[qn] (4.4)

with given q-hypergeometric term G(n), then L(n) = F(n)/G(n) is in F(qn). To solve
equation (4.4) we divide it by G(n) and search for all rational solutions L(n) of

J∑
j=0

(
σj(n)

j−1∏
i=0

γ(n+ i)
)

L(n+ j) = 1, γ(n) =
G(n+ 1)

G(n)
∈ F(qn) . (4.5)

Abramov (1995) showed how to determine a multiple of the denominator of L(n), there-
fore leaving the easier problem of finding polynomial solutions of inhomogeneous recur-
rence equations.‡

As an example, we search for all q-hypergeometric solutions F(k) of the inhomogeneous
recurrence equation

qk (q + 1) (q − 1) (qk+1 − 1) F(k + 2) + (q + 1) (1− qk+1) (qk (a q − a) + a− 1)

×F(k + 1) + (a+ 1) (a− 1) (qk+1 − 1) F(k) =
(qk + 1) (qk − 1) (a; q)k

(q; q)k
by the computation

> qrecsolve(q^k*(q+1)*(q-1)*(q^(k+1)-1)*F(k+2)+(q+1)*
(1-q^(k+1))*(q^k*(a*q-a)+a-1)*F(k+1)+(a+1)*(a-1)*(q^(k+1)-1)*F(k)=
(q^k+1)*(q^k-1)*qpochhammer(a,q,k)/qpochhammer(q,q,k),q,F(k));

[(
1− qk

)
qpochhammer(a, q, k)

qpochhammer(q, q, k)

]

4.3. power series solutions

Let Fq-hyp.[[qn]] denote the subset of the ring F[[qn]] of formal power series over F
whose coefficients form a q-hypergeometric sequence. Abramov et al. (1998) showed how
to search for solutions F(n) ∈ Fq-hyp.[[qn]] of

J∑
j=0

σj(n) F(n+ j) =
∞∑
k=0

bk
(
qn
)k
, (4.6)

where bk is a q-hypergeometric term w. r. t. k, based on the following lemma:

‡ To determine polynomial solutions one proceeds as described for the q-Gosper algorithm.



796 H. Böing and W. Koepf

Lemma 4.2. Assume F(n) =
∑∞
k=0 fk (qn)k ∈ Fq-hyp.[[qn]] is a solution of equation (4.6)

with

σj(n) =
δ∑
i=0

sj,i
(
qn
)i ∈ F[qn] , j = 0, . . . , J,

and δ = max
{

deg(σj(n))
∣∣ j = 0, . . . , J

}
. Then the recurrence equation

δ∑
i=0

( J∑
j=0

sj,δ−i q
i j
(
ql
)j)

fl+i = bl+δ, (4.7)

is valid for fl with l ∈ N0 and
l∑
i=0

( J∑
j=0

sj,l−i q
i j

)
fi = bl, l = 0, 1, . . . , δ − 1. (4.8)

Thus we can determine all solutions F(n) ∈ Fq-hyp.[[qn]] of equation (4.6) by using Lemma
4.2 and applying q-Petkovšek’s algorithm if the right-hand side is zero, and otherwise
Abramov’s algorithm to determine the coefficients fl via (4.7). Then we have to check if
we can adjust the solution for fl so that the initial conditions (4.8) are fulfilled.

With the option solution=series, the procedure qrecsolve searches for formal power
series solutions by the above algorithm. For example, given the recurrence equation†

q2 q2n F(n+ 3) + (1 + q) qn F(n+ 2) + (1− qn) F(n+ 1)− F(n) = 0,

we obtain the solution:

> qrecsolve(q^2*q^(2*n)*F(n+3)+(1+q)*q^n*F(n+2)+(1-q^n)*F(n+1)-F(n),
q,F(n),solution=series);

[ ∞∑
i=0

C1 q i2
(
qn
) i

qpochhammer(q, q, i)

]
.

Now we want to use this method to introduce an extension of q-Gosper’s algorithm, by
considering Example (2.9) again. First we compute polynomials P(k),Q(k),R(k) accord-
ing to (2.3) for Fn(k) defined by (2.9)

P(k) = 1, Q(k) = a qk − 1, and R(k) = a qn
(
qk+1 − 1

)
.

The recurrence equation for X(k) (see (2.5)) is then(
a qk − 1

)
X(k) − a qn

(
qk − 1

)
X(k − 1) = 1.

We know already that there is no Laurent polynomial solution X(k). However, computing
a degree bound for X(k) shows that X(k) could be a polynomial of degree n. Thus we
might try to look for a solution X(k) ∈ Fq-hyp.

[[
qk
]]

:

> qrecsolve((a*q^k-1)*X(k)-a*q^n*(q^k-1)*X(k-1)=1,q,X(k),solution=series);

† Abramov et al. (1998, Section 5.3, Example 8).

 ∞∑
i=0

 qpochhammer
(
q−n, q, i

)
q i
(
qk
) i

(
a qn − q i

)
qpochhammer

(
q−n

a
, q, i

)

 .

Note that (q−n; q) i = 0 for i > n, i.e. X(k) is a polynomial in qk of degree n as
expected. With equation (2.6) we can now build one antidifference G(k) for all n ∈ N0.
Our procedure qgosper can do all the necessary calculations:

> assume(n,posint);
> qgosper(qpochhammer(a,q,k)/qpochhammer(q,q,k)/(a*q^n)^k,q,k,series=true);

(
− 1 + qk

)
a qn qpochhammer(a, q, k)(

a qn
)k qpochhammer(q, q, k)

n∑
i=0

 qpochhammer
(
q−n, q, i

)
q i
(
qk
) i

(
a qn − q i

)
qpochhammer

(
q−n

a
, q, i

)
.

Note that after a second qgosper delivers an antidifference for all n ∈ N. This should be
compared with the timings in Table 1.

4.4. description of the Maple procedure qrecsolve

Calling Sequence:

qrecsolve(RE,q,f(n),...);

Parameters:
RE — an algebraic expression or an equation, the recurrence equa-

tion for f(n)
q — a name
f — a name, the recurrence function
n — a name, the recurrence variable
... — a sequence of optional arguments (of type name=name)

Input/Output Description:

This procedure decides whether a given recurrence equation with polynomial co-
efficients in qn has any q-hypergeometric solutions f(n). If no solution exists, an
empty list is returned. Otherwise a list is returned, where each solution is stored in
another list.

Options and Remarks:

• Using the option solution with one of the possible values polynomial, rational,
qhypergeometric, or series tells qrecsolve to search only for this type of solu-
tions. When set to series, qrecsolve returns all formal power series solutions of

the form f(n) =
∞∑
k=0

ak (qn)k. The default is to search for q-hypergeometric solutions.

• With the option return one can control the way solutions are returned. It may
be set to downratio, upratio, downrec, uprec, or qhypergeometric; the default
is uprec, meaning that an upward recurrence equation of first order is returned
for each possible solution f(n). Using the value downratio, qrecsolve will return
the ratio f(n)/f(n − 1). With return=qhypergeometric the procedure will return
a list of lists where each sublist consists of two entries. The first element is a q-
hypergeometric solution f(n) of the recurrence equation and the second element
contains restrictions on n.

• The option split controls to some extent the introduction of additional roots which
are required to solve the recurrence equation. The leading and trailing coefficient
of the recurrence equation have to be factored completely in order to obtain all
solutions. With split set to false only rational factorization is used. Thus by
specifying split=false the procedure might run faster but some solutions—which
require the introduction of new roots—may be lost. By default it is set to true.

5. Conclusion

Our package qsum.mpl is an efficient implementation of the q -Gosper, q-Zeilberger,
q-Petkovšek and similar algorithms. It is the first package which combines all these al-
gorithms which are useful tools to deal with problems associated with q-hypergeometric
series, in particular with q-analogues of orthogonal polynomials. Particularly, the com-
bination of q-Zeilberger with q-Petkovšek can be used to decide whether a definite q-
hypergeometric sum has a representation as a q-hypergeometric term.

